

# CS4012 Topics in Functional Programming

Glenn Strong

# Concurrency and Software Transactional Memory

## Concurrent Haskell

- We have looked at parallel programming
- There is a separate but related approach: Concurrency
- (possibly) non-deterministic
- Explicitly threaded with inter-thread communication
- Control.Concurrent

The basic primitives are very simple. To spawn a new thread of execution:

```
1 forkIO :: IO () -> IO ThreadId
1 main = do
2   forkIO (forever $ putChar 'o')
3   forkIO (forever $ putChar '0')
```

- You can use the thread ID to check the status of a thread, send it an exception, etc.
- These are not OS threads; GHC threads are lightweight and it's perfectly practical to have thousands of them at once.

Basic thread functions:

```
1 forkIO :: IO () -> IO ThreadId
2 killThread :: ThreadId -> IO ()
3 threadDelay :: Int -> IO ()
```

The program ends when the original (main) thread ends (this is called demonic concurrency)!

The previous example wasn't actually right. It should be:

```
1 main = do
2   forkIO (forever $ putChar 'o')
3   forkIO (forever $ putChar '0')
4   threadDelay (106)
```

We need some way to communicate between threads. One simple approach is to use a “channel” (a kind of unbounded FIFO)

You can write to the channel whenever you want

Reads from the channel block until something is available

```
1 newChan :: IO (Chan a)
2
3 writeChan :: Chan a -> a -> IO ()
```

```

4 readChan :: Chan a -> IO a
5
6 getChanContents :: Chan a -> IO [a]
7
8 isEmptyChan :: Chan a -> IO Bool
9 dupChan :: Chan a -> IO (Chan a)

```

Threads can communicate through channels.

```

1 main = do
2   hSetBuffering stdout NoBuffering
3   c <- newChan
4   forkIO (worker c)
5   forkIO (forever $ putChar '*')
6   readChan c
7
8 worker :: Chan Bool -> IO ()
9 worker c = do
10  mapM putChar "Printing all the chars"
11  writeChan c True

```

```
Pr***i*n*t*i*n*g* *a*l*l* *t*h*e* *c*h*a*r*s*****
```

Channels provide a nice abstraction that allows threads to communicate effectively – but watch out, you can introduce deadlocks using them.

Typically you might use them in:

- Servers (where you fork a new thread for each connection)
- Background processes (where the data computed by a thread becomes available incrementally)
- etc...

Channels are the communication primitives in languages like Erlang and Go.

In Haskell they are actually built out of something simpler

The basic communication primitive in Haskell is something called an `MVar`.

```

1 newEmptyMVar :: IO (MVar a)
2 takeMVar :: MVar a -> IO a
3 putMVar :: MVar a -> a -> IO ()

```

- An `MVar` can be either empty or full.
- `takeMVar` will block when it's empty
- `putMVar` will block when it's full

Simple use:

```

1 main = do
2   m <- newEmptyMVar
3   forkIO (do putMVar m 'a'; putMVar m 'b')

```

```

4   c <- takeMVar m
5   print c
6   c <- takeMVar m
7   print c

```

- We can use an MVar as a mutex for some other shared state
- Or as a one-item channel
- Or for shared state
- Or to build larger abstractions (like Chan).

But building larger abstractions from them can be tricky

If you look at how to build channels you can see the problem

You might want to say:

```

1  data Chan a = MVar [a]

```

(Example from Simon Marlow’s discussion of Channels)


Think about how readChan needs to block when the channel is empty

So instead we can build it as a sort of linked-list of MVar’s

```

1  type Stream a = MVar (Item a)
2  data Item a = Item a (Stream a)
3  data Chan a = Chan (MVar (Stream a)) -- read pointer
4                  (MVar (Stream a)) -- write pointer

```




An empty channel contains only the “hole” into which the first item in the stream will be placed.

```

1  newChan :: IO (Chan a)
2  newChan = do
3    hole <- newEmptyMVar
4    readVar <- newMVar hole
5    writeVar <- newMVar hole
6    return (Chan readVar writeVar)


```


The hole itself is represented by an MVar, as are the two end pointers for the linked list.



Adding an item is quite straightforward

```
1 writeChan :: Chan a -> a -> IO ()
2 writeChan (Chan _ writeVar) val = do
3   newhole <- newEmptyMVar
4   oldhole <- takeMVar writeVar
5   putMVar oldhole (Item val newhole)
6   putMVar writeVar newhole
```






```

writeChan (Chan = writeVar val) = do
  oldhole ← takeMVar writeVar
  newhole ← newEmpty MVar
  putMVar oldhole (IItem val newhole)
  getMVar writeVar newhole


```



```

writeChan (Chan = writeVar val) = do
  oldhole ← takeMVar writeVar
  newhole ← newEmpty MVar
  putMVar oldhole (IItem val newhole)
  getMVar writeVar newhole



```



```

writeChan (Chan = writeVar val) = do
  oldhole ← takeMVar writeVar
  newhole ← newEmpty MVar
  putMVar oldhole (IItem val newhole)
  getMVar writeVar newhole

```



Removing a value (“reading”) from the channel is similar:

```

1 readChan :: Chan a -> IO a
2 readChan (Chan readVar) = do
3   stream <- takeMVar readVar
4   Item val new <- takeMVar stream
5   putMVar readVar new
6   return val

```

If we think about how the channel blocks on reading we see that we can use this implementation to create *multicast* channels:

```

1 dupChan :: Chan a -> IO (Chan a)
2 dupChan (Chan _ writeVar) = do
3   hole <- takeMVar writeVar
4   putMVar writeVar hole
5   newReadVar <- newMVar hole
6   return (Chan newReadVar writeVar)

```

This operation leaves us with two channels which share their “write” pointer, but have separate “read” pointers.

This definition of `dupChan` will interact badly with our implementation of `readChan`, since `readChan` didn't originally need to return the value to the "hole".

Using the standard `Control.Concurrent` operation `readMVar` (defined thus):

```
1 readMVar :: MVar a -> IO a
2 readMVar m = do
3   a <- takeMVar m
4   putMVar m a
5   return a
```

We can fix `readChan` so that it plays nicely with `dupChan`:

```
1 readChan :: Chan a -> IO a
2 readChan (Chan readVar _) = do
3   stream <- takeMVar readVar
4   Item val tail <- readMVar stream
5   putMVar readVar tail
```

Lastly, we might be tempted to implement a form of "peeking" for channels:

```
1 unGetChan :: Chan a -> a -> IO ()
2 unGetChan (Chan readVar ) val = do
3   newReadEnd <- newEmptyMVar
4   readEnd <- takeMVar readVar
5   putMVar newReadEnd (Item val readEnd)
6   putMVar readVar newReadEnd
```

- This is superficially OK, but...
- Consider the case "peeking" at an empty channel
  - thread 1 reads from the channel
  - and thread 2 attempts to do an `unGetChan`

## Software Transactional Memory

- When we talk about co-ordination in shared-memory concurrent programs Locks and Condition variables are the standard technique
- Locks are ridiculously easy to get wrong - Races (when we forget to lock) - Deadlocks (when we lock/release in the wrong order) - Error recovery is hard (e.g. corruption from uncaught exceptions)
- And they are not compositional - We can't build a working system from working pieces

A program with a small number of big locks is usually manageable, but can expect a lot of blocked threads



We could add more granular locks, but then it gets hard to keep the program correct.

Quick example<sup>1</sup>

```
1 transferFunds a1 a2 amount = do
2   withdraw a1 amount
3   deposit a2 amount
```

Another thread could observe a time when

- The money has left the first account
- But not arrived in the second account

If we try to be clever:

```
1 transferFunds a1 a2 amount = do
2   lock a1; lock a2
3   withdraw a1 amount
4   deposit a2 amount
5   unlock a2; unlock a1
```

then we might deadlock.

- The whole thing is a headache, totally non-modular
- Often we end up wishing we had never heard of this concurrency business.

Need a better idea.

- Recently (mainly post 2005) one has been getting some attention
- Software Transactional Memory
- Steal the idea of “Transactions” from database people
  - This means computations can be done atomically
  - Mix in the idea of pure, first-class functions

In a nutshell:

```
1 atomically :: Int -> Account -> Account -> IO ()
2 transferFunds a1 a2 amount = atomically $ do
3   withdraw a1 amount
4   deposit a2 amount
```

- The atomic block commits in an all-or-nothing way
- Executes in isolation
- Cannot deadlock, can generate exceptions
- One possible execution strategy (“optimistic concurrency”):
  - Execute code lock-free,
  - logging all memory access instead of performing it
  - At the end, commit the log, retrying the block on failure

---

<sup>1</sup>See “Beautiful Concurrency”, link on BlackBoard

OK, but:

- We must not touch any of the transaction variables outside an atomic block
- We must not have any side-effects inside an atomic block

Type system to the rescue.

```
atomically :: STM a -> IO a
```

- The STM monad actions have side-effects but far more limited ones than IO
- Mainly they are about reading and writing special transaction variables

```
atomically :: STM a -> IO a
```

```
newTVar :: a -> STM (TVar a)
readTVar :: TVar a -> STM a
writeTVar :: TVar a -> a -> STM ()
```

- Semantically TVar values are just containers for values
  - in particular they don't have the blocking semantics of MVar's

We can implement withdraw using this.

```
1 type Account = TVar Int
2
3 withdraw :: Account -> Int -> STM ()
4 withdraw acc amount = do
5   bal <- readTVar acc
6   writeTVar acc (bal - amount)
```

The type system keeps us honest:

```
1 bad :: Account -> IO ()
2 bad acc = do print "Withdrawing..."
3 withdraw acc 10
```

- Won't even compile! withdraw is an STM action.

We can fix it:

```
1 bad acc = do print "Withdrawing..."
2   atomically $ withdraw acc 10
```

- This satisfies the first condition (must not touch the affected variables outside the atomic block.

STM has some other new ideas:

```
1 retry :: STM ()
```

It means “abandon and re-execute from the start”

Implementation will block on all read variables before retrying

```

1 withdraw :: Account -> Int -> STM ()
2 withdraw acc amount = do
3   bal <- readTVar acc
4   if bal < amount then retry
5   else writeTVar acc (bal - amount)

```

We cannot nest uses of atomically (what would that even mean?)

STM offers “compositional choice” which covers a lot of the real cases where we might try that:

```

1 orElse :: Stm a -> Stm a -> Stm a
2
3 atomically $ do withdraw a1 x `orElse` withdraw a2 x
4               deposit a3 x

```

Since we are tracking TVars we can even include the idea of *invariants*

```
always :: STM Bool -> STM ()
```

Which we can use to maintain a (global) pool of invariants that are checked after each transaction. Transactions that break the invariants are retried

```

1 newAccount = do
2   v <- newTVar 0
3   always $ do cts <- readTVar v ; return (cts >= 0)
4   return v

```

Finally, it’s probably obvious why we must not allow IO actions inside an atomic block?

```

1 atomically $ do
2   x <- readTVar xv
3   y <- readTVar yv
4   if x>y then launchMissiles
5   else return ()

```

Fortunately the type checker won’t allow this!

```
No instance for (Control.Monad.IO.Class.MonadIO STM)
arising from a use of 'liftIO'
```

- Versions of STM have been developed (and are used) for C++, Java, C#
- Some difficult to solve problems, though:
  - Retry semantics in impure languages
  - Keeping atomic blocks IO free
  - Keeping the transaction variables untouched
  - Keeping the atomic blocks away from normal variables

## Channels in STM

Finally, let's consider how to construct communication channels, as an example of how STM simplifies the concurrency model.

(this implementation is available in `Control.Concurrency.STM.TChan`)

The interface is simple

```
1 data TChan a
2
3 newTChan    :: STM (TChan a)
4 writeTChan :: TChan a -> a -> STM ()
5 readTChan  :: TChan a -> STM a
```

We can implement using the same linked-list model we used for MVar based channels:

```
1 data TChan a = TChan (TVar (TVarList a))
2                  (TVar (TVarList a))
3
4 type TVarList a = TVar (TList a)
5 data TList a = TNil | TCons a (TVarList a)
```

(we have an explicit TNil for the empty list. This was represented by an empty MVar before).

```
1 newTChan :: STM (TChan a)
2 newTChan = do
3   hole <- newTVar TNil
4   read <- newTVar hole
5   write <- newTVar hole
6   return (TChan read write)
7
8 readTChan :: TChan a -> STM a
9 readTChan (TChan readVar _) = do
10  listHead <- readTVar readVar
11  head <- readTVar listHead
12  case head of
13    TNil      -> retry
14    TCons val tail -> do
15      writeTVar readVar tail
16      return val
```

Notice how blocking in a read is implemented using `retry`

```
1 writeTChan :: TChan a -> a -> STM ()
2 writeTChan (TChan _ writeVar) a = do
3   newListEnd <- newTVar TNil
4   listEnd <- readTVar writeVar
```

```

5   writeTVar writeVar newListEnd
6   writeTVar listEnd (TCons a newListEnd)

```

Remember how stuck we were trying to implement `unGetChan`?

The `MVar` semantics meant we could never be sure that our program wouldn't deadlock if some thread was using this operation.

```

1  unGetTChan :: TChan a -> a -> STM ()
2  unGetTChan (TChan readVar _) a = do
3    listHead <- readTVar readVar
4    newHead <- newTVar (TCons a listHead)
5    writeTVar readVar newHead

```

There are other operations possible too:

```

1  isEmptyTChan :: TChan a -> STM Bool
2  isEmptyTChan (TChan read _write) = do
3    listhead <- readTVar read
4    head <- readTVar listhead
5    case head of
6      TNil      -> return True
7      TCons _ _ -> return False

```

Exercise: can this be (safely) implemented in the `MVar` version of channels?

How about this:

```

1  readEitherTChan :: TChan a -> TChan b -> STM (Either a b)
2  readEitherTChan a b =
3    fmap Left (readTChan a)
4    `orElse`
5    fmap Right (readTChan b)

```

## Summary

- STM is a promising alternative to lock-based synchronization for concurrent programming
- it offers *composable*
  - atomicity
  - blocking
  - error handling

There is a price, of course, and it's worth noting it:

- `MVar` based concurrency is generally faster
  - but you can sometimes use STM to construct simpler solutions that will be faster than an `MVar` based solution.
- `MVar` based solutions can guarantee *fairness*

- Multiple `MVar` blocked threads will be woken in FIFO order.
- We can't be *fair* to `TVar` threads unless we abandon composability

The GHC implementation of STM is very efficient as it is tightly bound into the RTS.

- Each transaction maintains a thread-local *transaction log* recording reads and (tentative) writes.
- Reads consult the log, writes are made only to the log
- When attempting to commit an atomic block the log is checked for validity
  - If the old value of a read variable is pointer-equal to the current `TVar` value then it passes the validity check
- If all log entries pass then the changes can be applied; this is done without interruption.
- If any validations fail then the log is discarded and a fresh transaction is started with a new log.
- When a `retry` is required the log is discarded and the transaction is added to a wait-queue associated with the `TVars` in the log
- Any commit that updates the relevant `TVars` will be unblocked and scheduled to be run again.

Fully portable software-based implementations also exist (see, for example, “A Haskell-Implementation of STM Haskell with Early Conflict Detection” by David Sabel).

- Portable
- But significantly slower than the GHC internal implementation